

Chapter 6: MSI Logic

This chapter introduces a group of combinational logic building blocks that are commonly used in digital design. As we move into systems that are larger than individual gates, there are naming conventions that are used to describe the size of the logic. Table 6.1 gives these naming conventions. In this chapter we look at *medium-scale integrated circuit* (MSI) logic. Each of these building blocks can be implemented using the combinational logic design steps covered in Chaps. 4 and 5. The goal of this chapter is to provide an understanding of the basic principles of MSI logic.

Commonly Used Names to Describe The Size of Digital Logic		
Name	Example	# of Transistors
SSI - Small Scale Integrated Circuits	Individual Gates (NAND, INV)	10's
MSI - Medium Scale Integrated Circuits	Decoders, Multiplexers	100's
LSI - Large Scale Integrated Circuits	Arithmetic Circuits, RAM	1k - 10k
VLSI - Very Large Scale Integrated Circuits	Microprocessors	100k - 1M

While there are names for logic sizes above 1M transistor such as ULSI (Ultra), the term "VLSI" is now used to describe all integrated circuits that are so large they require CAD tools for their design, synthesis and implementation.

Table 6.1
Naming convention for the size of digital systems

Learning Outcomes—After completing this chapter, you will be able to:

- 6.1 Design a decoder circuit using both the classical digital design approach and the modern HDL-based approach.
- 6.2 Design an encoder circuit using both the classical digital design approach and the modern HDL-based approach.
- 6.3 Design a multiplexer circuit using both the classical digital design approach and the modern HDL-based approach.
- 6.4 Design a demultiplexer circuit using both the classical digital design approach and the modern HDL-based approach.

6.1 Decoders

A decoder is a circuit that takes in a binary *code* and has outputs that are asserted for specific values of that code. The code can be of any type or size (e.g., unsigned, two's complement). Each output will assert for only specific input codes. Since combinational logic circuits only produce a single output, this means that within a decoder, there will be a separate combinational logic circuit for each output.

6.1.1 Example: One-Hot Decoder

A one-hot decoder is a circuit that has n inputs and 2^n outputs. Each output will assert for one and only one input code. Since there are 2^n outputs, there will always be one and only one output asserted at

any given time. Example 6.1 shows the process of designing a 2-to-4 one-hot decoder by hand (i.e., using the classical digital design approach).

Example: 2-to-4 One-Hot Decoder - Logic Synthesis by Hand

The block diagram and truth table for this system are as follows:

A	B	F3	F2	F1	F0
0	0	0	0	0	1
0	1	0	0	1	0
1	0	0	1	0	0
1	1	1	0	0	0

Each output asserts for a specific input code. This is where the term "one-hot" comes from. Each output is only "hot" for one input code.

When designing this circuit, each output needs to have its own separate combinational logic circuit. This is the same as if there were four separate truth tables. This design could be implemented using 4x, 2-input K-maps to form the logic expressions for these outputs; however, by inspection a minterm list for each output will be the most minimal circuit.

$$F0 = \sum_{A,B}(0) = A'B'$$

$$F2 = \sum_{A,B}(2) = A'B$$

$$F1 = \sum_{A,B}(1) = A'B$$

$$F3 = \sum_{A,B}(3) = A \cdot B$$

When implementing the final decoder, the input inversions for A and B can be shared across all of the AND gates.


Timing Waveform

Example 6.1
2-to-4 One-Hot Decoder—Logic Synthesis by Hand

As decoders get larger, it is necessary to use hardware description languages to model their behavior. Example 6.2 shows how to model a 3-to-8 one-hot decoder in VHDL with concurrent signal assignments and logic operators.

Example: 3-to-8 One-Hot Decoder – VHDL Modeling using Logical Operators

The block diagram and truth table for this system are as follows:



To implement this in VHDL using logical operators, we must first determine the logic that will be used in the concurrent signal assignments. Again, since each logic function only has one input code corresponding to an output of '1', the minterm can be used to implement the logic.

$$\begin{aligned}
 F0 &= \sum_{A,B,C}(0) = A'B'C' & F4 &= \sum_{A,B,C}(4) = A'B'C \\
 F1 &= \sum_{A,B,C}(1) = A'B'C & F5 &= \sum_{A,B,C}(5) = A'B \cdot C \\
 F2 &= \sum_{A,B,C}(2) = A'B \cdot C' & F6 &= \sum_{A,B,C}(6) = A \cdot B \cdot C' \\
 F3 &= \sum_{A,B,C}(3) = A'B \cdot C & F7 &= \sum_{A,B,C}(7) = A \cdot B \cdot C
 \end{aligned}$$

In VHDL, each of the outputs requires a separate signal assignment.

```

entity decoder_1hot_3to8 is
  port (A,B,C          : in bit;
        F0,F1,F2,F3,F4,F5,F6,F7 : out bit);
end entity;

architecture decoder_1hot_3to8_arch of decoder_1hot_3to8 is
begin
  F0 <= (not A) and (not B) and (not C);
  F1 <= (not A) and (not B) and (C);
  F2 <= (not A) and (B) and (not C);
  F3 <= (not A) and (B) and (C);
  F4 <= (A) and (not B) and (not C);
  F5 <= (A) and (not B) and (C);
  F6 <= (A) and (B) and (not C);
  F7 <= (A) and (B) and (C);
end architecture;


```

Example 6.2 3-to-8 One-Hot Decoder—VHDL Modeling Using Logical Operators

This description can be further simplified by using vector notation for the ports and describing the functionality using either conditional or select signal assignment. Example 6.3 shows how to model the 3-to-8 one-hot decoder in VHDL with conditional and select signal assignments.

Example: 3-to-8 One-Hot Decoder – VHDL Modeling – Conditional/Select Signal Assignments

The block diagram and truth table for this system are as follows. Notice that the input and output ports now use type `bit_vector` in order to create a more compact description.



ABC	F(7)	F(6)	F(5)	F(4)	F(3)	F(2)	F(1)	F(0)
"000"	0	0	0	0	0	0	0	1
"001"	0	0	0	0	0	0	1	0
"010"	0	0	0	0	0	1	0	0
"011"	0	0	0	0	1	0	0	0
"100"	0	0	0	1	0	0	0	0
"101"	0	0	1	0	0	0	0	0
"110"	0	1	0	0	0	0	0	0
"111"	1	0	0	0	0	0	0	0

The following is the entity for this design that uses `bit_vectors` for the inputs and outputs.

```
entity decoder_1hot_3to8 is
  port (ABC : in bit_vector(2 downto 0);
        F   : out bit_vector(7 downto 0));
end entity;
```

The following are two different ways to implement the behavior of the decoder: (1) conditional signal assignments and (2) selected signal assignments. In these techniques the signal assignment can be made to the entire `F` vector instead of to the individual scalar outputs. This creates a compact VHDL model but will still synthesis into eight separate combinational logic circuits.

```
(1) architecture decoder_1hot_3to8_arch of decoder_1hot_3to8 is
begin
  F <= "00000001" when (ABC = "000") else
       "00000010" when (ABC = "001") else
       "00000100" when (ABC = "010") else
       "00001000" when (ABC = "011") else
       "00010000" when (ABC = "100") else
       "00100000" when (ABC = "101") else
       "01000000" when (ABC = "110") else
       "10000000" when (ABC = "111");
end architecture;
```

```
(2) architecture decoder_1hot_3to8_arch of decoder_1hot_3to8 is
begin
  with (ABC) select
    F <= "00000001" when "000",
         "00000010" when "001",
         "00000100" when "010",
         "00001000" when "011",
         "00010000" when "100",
         "00100000" when "101",
         "01000000" when "110",
         "10000000" when "111";
end architecture;
```

Example 6.3


3-to-8 One-Hot Decoder—VHDL Modeling Using Conditional and Select Signal Assignments

6.1.2 Example: Seven-Segment Display Decoder

A seven-segment display decoder is a circuit used to drive character displays that are commonly found in applications such as digital clocks and household appliances. A character display is made up of seven individual LEDs, typically labeled a–g. The input to the decoder is the binary equivalent of the decimal or Hex character that is to be displayed. The output of the decoder is the arrangement of LEDs that will form the character. Decoders with 2-inputs can drive characters “0” to “3.” Decoders with 3-inputs can drive characters “0” to “7.” Decoders with 4-inputs can drive characters “0” to “F” with the case of the Hex characters being “A, b, c or C, d, E, and F.”

Let’s look at an example of how to design a 3-input, seven-segment decoder by hand. The first step in the process is to create the truth table for the outputs that will drive the LEDs in the display. We’ll call these outputs F_a, F_b, \dots, F_g . Example 6.4 shows how to construct the truth table for the seven-segment display decoder. In this table, a logic 1 corresponds to the LED being ON.

Example: 7-Segment Display Decoder - Truth Table




A	B	C	F_a	F_b	F_c	F_d	F_e	F_f	F_g
0	0	0	1	1	1	1	1	1	0
0	0	1	0	1	1	0	0	0	0
0	1	0	1	1	0	1	1	0	1
0	1	1	1	1	1	1	0	0	1
1	0	0	0	1	1	0	0	1	1
1	0	1	1	0	1	1	0	1	1
1	1	0	1	0	1	1	1	1	1
1	1	1	1	1	1	0	0	0	0

Example 6.4
Seven-Segment Display Decoder—Truth Table



If we wish to design this decoder by hand we need to create seven separate combinational logic circuits. Each of the outputs (F_a – F_g) can be put into a 3-input K-map to find the minimized logic expression. Example 6.5 shows the design of the decoder from the truth table in Example 6.4 by hand.

Example: 7-Segment Display Decoder – Logic Synthesis by Hand

The block diagram and truth table for this system are as follows:



Each output of the decoder needs its own logic expression.




Example 6.5
Seven-Segment Display Decoder—Logic Synthesis by Hand

This same functionality can be modeled in VHDL using concurrent signal assignments with logical operators. Example 6.6 shows how to model the seven-segment decoder in VHDL using concurrent signal assignments with logic operators.

Example: 7-Segment Display Decoder – VHDL Modeling using Logical Operators

The block diagram and truth table for this system are as follows:



```
entity decoder_7seg is
  port (A,B,C          : in bit;
        Fa,Fb,Fc,Fd,Fe,Fg,Fg : out bit);
end entity;


architecture decoder_7seg_arch of decoder_7seg is
begin
  Fa <= ((not A) and (not C)) or B or (A and C);
  Fb <= ((not B) and (not C)) or (not A) or (B and C);
  Fc <= A or (not B) or C;
  Fd <= ((not A) and (not C)) or ((not A) and B) or (B and (not C))
        or (A and (not B) and C);
  Fe <= ((not A) and (not C)) or (B and (not C));
  Ff <= ((not B) and (not C)) or (A and (not C)) or (A and (not B));
  Fg <= ((not A) and B) or (A and (not C)) or (A and (not B));
end architecture;
```

Example 6.6 Seven-Segment Display Decoder—Modeling Using Logical Operators

Again, a more compact description of the decoder can be accomplished if the ports are described as vectors and a conditional or select signal assignment is used. Example 6.7 shows how to model the seven-segment decoder in VHDL using conditional and selected signal assignments.

Example: 7-Segment Decoder – VHDL Modeling – Conditional/Select Signal Assignments

The block diagram and truth table for this system are as follows:



The following is the entity for this design that uses bit_vectors for the inputs and outputs.

```
entity decoder_7seg is
  port (ABC : in bit_vector(2 downto 0);
        F   : out bit_vector(6 downto 0));
end entity;
```

The following are two different ways to implement the behavior of the decoder: (1) conditional signal assignments and (2) selected signal assignments.

```
(1) architecture decoder_7seg_arch of decoder_7seg is
  begin
    F <= "1111110" when (ABC = "000") else
         "0110000" when (ABC = "001") else
         "1101101" when (ABC = "010") else
         "1111001" when (ABC = "011") else
         "0110011" when (ABC = "100") else
         "1011011" when (ABC = "101") else
         "1011111" when (ABC = "110") else
         "1110000" when (ABC = "111");
  end architecture;
```

```
(2) architecture decoder_7seg_arch of decoder_7seg is
  begin
    with (ABC) select
      F <= "1111110" when "000",
          "0110000" when "001",
          "1101101" when "010",
          "1111001" when "011",
          "0110011" when "100",
          "1011011" when "101",
          "1011111" when "110",
          "1110000" when "111";
  end architecture;
```

Example 6.7

Seven-Segment Display Decoder—Modeling Using Conditional and Selected Signal Assignments

CONCEPT CHECK

CC6.1 In a decoder, a logic expression is created for each output. Once all of the output logic expressions are found, how can the decoder logic be further minimized?

- By using K-maps to find the output logic expressions.
- By buffering the inputs so that they can drive a large number of other gates.
- By identifying any logic terms that are used in multiple locations (inversions, product terms, and sum terms) and sharing the interim results among multiple circuits in the decoder.
- By ignoring fan-out.

6.2 Encoders

An encoder works in the opposite manner as a decoder. An assertion on a specific input port corresponds to a unique code on the output port.

6.2.1 Example: One-Hot Binary Encoder

A one-hot binary encoder has n outputs and 2^n inputs. The output will be an n -bit, binary code which corresponds to an assertion on one and only one of the inputs. Example 6.8 shows the process of designing a 4-to-2 binary encoder by hand (i.e., using the classical digital design approach).

Example: 4-to-2 Binary Encoder – Logic Synthesis by Hand
 The block diagram and truth table for this system are as follows:

A	B	C	D	Y	Z
0	0	0	1	0	0
0	0	1	0	0	1
0	1	0	0	1	0
1	0	0	0	1	1

D => "00"
 C => "01"
 B => "10"
 A => "11"

When designing this circuit, each output needs to have its own separate combinational logic circuit. When constructing the K-maps for Y and Z, each will have 4-inputs (A, B, C, D). The output values for many of the input codes are not specified in the above truth table. As such, we can use Don't Cares (X) to simplify the logic.

Y

		AB			
		00	01	11	10
CD	00	X	1	X	1
	01	0	X	X	X
	11	X	X	X	X
	10	0	X	X	X

→ Y = A + B

Z

		AB			
		00	01	11	10
CD	00	X	0	X	1
	01	0	X	X	X
	11	X	X	X	X
	10	1	X	X	X

→ Z = A + C

decoder_1hot_2to4

Notice that D is not used.


Timing Waveform

Example 6.8
 4-to-2 Binary Encoder—Logic Synthesis by Hand

In VHDL this can be implemented using logical operators, conditional signal assignments, or selected signal assignments. In the conditional and selected signal assignments, if an output is not listed for each and every input possibility, then an output must be specified to cover any remaining input conditions. In the conditional signal assignment, the covering value is specified after the final *else* statement. In the selected signal assignment, the covering value is specified using the *when others* clause. Example 6.9 shows how to model the encoder in VHDL using each of the abovementioned modeling techniques.

Example: 4-to-2 Binary Encoder – VHDL Modeling

The block diagram and truth table for this system are as follows:



The following is the entity for this design that uses bit_vectors for the inputs and outputs.

```
entity encoder_1hot_4to2 is
    port (ABCD : in bit_vector(3 downto 0);
          YZ   : out bit_vector(1 downto 0));
end entity;
```

The following are three different ways to implement the behavior of the encoder: (1) concurrent signal assignments with logical operators; (2) conditional signal assignment; and (3) selected signal assignments.

```
(1) architecture encoder_1hot_4to2_arch of encoder_1hot_4to2 is
    begin
        YZ(1) <= ABCD(3) or ABCD(2);
        YZ(0) <= ABCD(3) or ABCD(1);
    end architecture;
```

```
(2) architecture encoder_1hot_4to2_arch of encoder_1hot_4to2 is
    begin
        YZ <= "00" when (ABCD = "0001") else
              "01" when (ABCD = "0010") else
              "10" when (ABCD = "0100") else
              "11" when (ABCD = "1000") else
              "00";
    end architecture;
```

```
(3) architecture encoder_1hot_4to2_arch of encoder_1hot_4to2 is
    begin
        with (ABCD) select
            YZ <= "00" when "0001",
                  "01" when "0010",
                  "10" when "0100",
                  "11" when "1000",
                  "00" when others;
    end architecture;
```

Example 6.9
4-to-2 Binary Encoder—VHDL Modeling

CONCEPT CHECK


- CC6.2** If it is desired to have the outputs of an encoder produce 0's for all input codes not defined in the truth table, can “don't cares” be used when deriving the minimized logic expressions? Why?
- No. Don't cares aren't used in encoders.
 - Yes. Don't cares can always be used in K-maps.
 - Yes. All that needs to be done is to treat each X as a 0 when forming the most minimal prime implicant.
 - No. Each cell in the K-map corresponding to an undefined input code needs to contain a 0 so don't cares are not applicable.

6.3 Multiplexers

A multiplexer is a circuit that passes one of its multiple inputs to a single output based on a select input. This can be thought of as a digital switch. The multiplexer has n select lines, 2^n inputs, and one output. Example 6.10 shows the process of designing a 2-to-1 multiplexer by hand (i.e., using the classical digital design approach).

Example: 2-to-1 Multiplexer – Logic Synthesis by Hand

The symbol and truth table for the 2-to-1 multiplexer are as follows:



In order to design the multiplexer, it is helpful to list all possible values for A, B and Sel in a truth table form.


Sel	A	B	F
0	0	0	0
0	0	1	0
0	1	0	1
0	1	1	1
1	0	0	0
1	0	1	1
1	1	0	0
1	1	1	1

When Sel=0, the output is A →


When Sel=1, the output is B →

Sel	A	B	00	01	11	10
0	0	1	0	0	0	0
1	0	1	1	1	1	1

$$F = \text{Sel}'A + \text{Sel}B$$



Timing Waveform




Example 6.10
2-to-1 Multiplexer—Logic Synthesis by Hand

Again, in VHDL a multiplexer can be implemented using different behavioral models. Let's look at the modeling of a 4-to-1 multiplexer in VHDL using logical operators, conditional signal assignments, and selected signal assignments. This multiplexer requires two select lines to address each of the four input lines. Each of the product terms in the multiplexer logic expression must include both select lines. The polarity of the select lines is chosen so that when an input is selected, its product term will allow the input to pass to the OR gate. In the VHDL implementation of the multiplexer using conditional and selected signal assignments, since every possible value of *Sel* is listed, it is not necessary to use a final *else* or *when others* clause. Example 6.11 shows the VHDL modeling of a 4-to-1 multiplexer.

Example: 4-to-1 Multiplexer – VHDL Modeling

The symbol and truth table for the 4-to-1 multiplexer are as follows:



The following is the entity for this design that uses type `bit_vector` for the select input.

```
entity mux_4to1 is
    port (A,B,C,D : in bit;
          Sel      : in bit_vector(1 downto 0);
          F        : out bit);
end entity;
```

The following are three different ways to implement the behavior of the mux: (1) concurrent signal assignments with logical operators; (2) conditional signal assignment; and (3) selected signal assignments.

```
(1) architecture mux_4to1_arch of mux_4to1 is
    begin
        F <= (A and not Sel(0) and not Sel(1)) or
            (B and not Sel(0) and Sel(1)) or
            (C and Sel(0) and not Sel(1)) or
            (D and Sel(0) and Sel(1));
    end architecture;
```

```
(2) architecture mux_4to1_arch of mux_4to1 is
    begin
        F <= A when (Sel = "00") else
            B when (Sel = "01") else
            C when (Sel = "10") else
            D when (Sel = "11");
    end architecture;
```

```
(3) architecture mux_4to1_arch of mux_4to1 is
    begin
        with (Sel) select
            F <= A when "00",
                B when "01",
                C when "10",
                D when "11";
    end architecture;
```

Example 6.11
4-to-1 Multiplexer—VHDL Modeling

CONCEPT CHECK

CC6.3 How are the product terms in a multiplexer based on the identity theorem?


- A) Only the select product term will pass its input to the final sum term. Since all of the unselected product terms output 0, the input will be passed through the sum term because anything OR'd with a 0 is itself.
- B) The select lines are complemented such that they activate only one OR gate.
- C) The select line inputs will produce 1's on the inputs of the selected product term. This allows the input signal to pass through the selected AND gate because anything AND'd with a 1 is itself.
- D) The select line inputs will produce 0's on the inputs of the selected sum term. This allows the input signal to pass through the selected OR gate because anything OR'd with a 0 is itself.

6.4 Demultiplexers

A demultiplexer works in a complementary fashion to a multiplexer. A demultiplexer has one input that is routed to one of its multiple outputs. The output that is active is dictated by a select input. A demux has n select lines that chooses to route the input to one of its 2^n outputs. When an output is not selected, it outputs a logic 0. Example 6.12 shows the process of designing a 1-to-2 demultiplexer by hand (i.e., using the classical digital design approach).

Example: 1-to-2 Demultiplexer – Logic Synthesis by Hand

The symbol and truth table for the 1-to-2 demultiplexer are as follows:



Sel	Y	Z
0	A	0
1	0	A

In order to design the demultiplexer, it is helpful to list all possible values for A and Sel and the corresponding outputs on Y and Z. A separate circuit is needed for both Y and Z.

When Sel=0,
the Y = A

Sel	A	Y	Z
0	0	0	0
0	1	1	0
1	0	0	0
1	1	0	1


When Sel=1,
the Y = B

Sel	Y
0	0
1	0


→ Y = Sel'·A

Sel	Z
0	0
1	1

→ Z = Sel·A



Timing Waveform



When Sel=0,
Y=A. Y=0
otherwise.


When Sel=1,
Z=A. Z=0
otherwise.

Example 6.12
1-to-2 Demultiplexer—Logic Synthesis by Hand

Again, in VHDL a demultiplexer can be implemented using different behavioral models. Example 6.13 shows the modeling of a 1-to-4 demultiplexer in VHDL using logical operators, conditional signal assignments, and selected signal assignments. This demultiplexer requires two select lines in order to choose between the four outputs.

Example: 1-to-4 Demultiplexer – VHDL Modeling

The symbol and truth table for the 1-to-4 demultiplexer are as follows:



Sel	W	X	Y	Z
"00"	A	0	0	0
"01"	0	A	0	0
"10"	0	0	A	0
"11"	0	0	0	A

The following is the entity for this design that uses type `bit_vector` for the select input.

```
entity demux_1to4 is
  port (A      : in bit;
        Sel    : in bit_vector(1 downto 0);
        W,X,Y,Z : out bit);
end entity;
```

The following shows the behavior of the demux using: (1) concurrent signal assignments with logical operators; (2) conditional signal assignment; and (3) selected signal assignments.

```
(1) architecture demux_1to4_arch of demux_1to4 is
  begin
    W <= A and not Sel(0) and not Sel(1);
    X <= A and not Sel(0) and Sel(1);
    Y <= A and Sel(0) and not Sel(1);
    Z <= A and Sel(0) and Sel(1);
  end architecture;
```

```
(2) architecture demux_1to4_arch of demux_1to4 is
  begin
    W <= A when (Sel = "00") else '0';
    X <= A when (Sel = "01") else '0';
    Y <= A when (Sel = "10") else '0';
    Z <= A when (Sel = "11") else '0';
  end architecture;
```

```
(3) architecture demux_1to4_arch of demux_1to4 is
  begin
    with (Sel) select
      W <= A when "00", '0' when others;

    with (Sel) select
      X <= A when "01", '0' when others;

    with (Sel) select
      Y <= A when "10", '0' when others;

    with (Sel) select
      Z <= A when "11", '0' when others;
  end architecture;
```

Example 6.13
1-to-4 Demultiplexer—VHDL Modeling

CONCEPT CHECK


CC6.4 How many select lines are needed in a 1-to-64 demultiplexer?

- A) 1 B) 4 C) 6 D) 64


Summary

- ❖ The term medium-scale integrated circuit (MSI) logic refers to a set of basic combinational logic circuits that implement simple, commonly used functions such as decoders, encoders, multiplexers, and demultiplexers. MSI logic can also include operations such as comparators and simple arithmetic circuits.
- ❖ While an MSI logic circuit may have multiple outputs, each output requires its own unique logic expression that is based on the system inputs.
- ❖ A decoder is a system that has a greater number of outputs than inputs. The behavior of each output is based on each unique input code.
- ❖ An encoder a system that has a greater number of inputs than outputs. A compressed output code is produced based on which input(s) lines are asserted.
- ❖ A multiplexer is a system that has one output and multiple inputs. At any given time, one and only one input is routed to the output based on the value on a set of *select lines*. For n select lines, a multiplexer can support 2^n inputs.
- ❖ A demultiplexer is a system that has one input and multiple outputs. The input is routed to one of the outputs depending on the value on a set of select lines. For n select lines, a demultiplexer can support 2^n outputs.
- ❖ HDLs are particularly useful for describing MSI logic due to their abstract modeling capability. Through the use of Boolean conditions and vector assignments, the behavior of MSI logic can be modeled in a compact and intuitive manner.

7-Segment Display Decoder



7-Segment Display Layout



A	B	C	D	F _a	F _b	F _c	F _d	F _e	F _f	F _g
0	0	0	0	1	0	0	0	0	0	0
0	0	0	1	1	0	0	0	0	0	0
0	0	1	0	1	0	0	0	0	0	0
0	0	1	1	1	0	0	0	0	0	0
<hr/>										
0	1	0	0	1	0	0	0	0	0	0
0	1	0	1	1	0	0	0	0	0	0
0	1	1	0	1	0	0	0	0	0	0
0	1	1	1	1	0	0	0	0	0	0
<hr/>										
1	0	0	0	1	0	0	0	0	0	0
1	0	0	1	1	0	0	0	0	0	0
1	0	1	0	1	0	0	0	0	0	0
1	0	1	1	1	0	0	0	0	0	0
<hr/>										
1	1	0	0	1	0	0	0	0	0	0
1	1	0	1	1	0	0	0	0	0	0
1	1	1	0	1	0	0	0	0	0	0
1	1	1	1	1	0	0	0	0	0	0

Fig. 6.3
7-Segment Display Decoder Truth Table

6.1.6 Design a VHDL model for a 4-input, seven-segment HEX character decoder using conditional signal assignments. Use the entity definition given in Fig. 6.4 for your design. The system has a 4-bit input vector called A and a 7-bit output vector called F. The individual scalars within the output vector (i.e., F(6 downto 0)) correspond to the character display segments a, b, c, d, e, f, and g, respectively. A logic 1 on an output corresponds to the LED being ON. The display will show the HEX characters 0–9, A, b, c, d, E, and F corresponding to the 4-bit input code on A. A

template for creating the truth table is provided in Fig. 6.3. The signals in this table correspond to the entity in this problem as follows: A = A(3), B = A(2), C = A(1), D = A(0), F_a = F(6), F_b = F(5), F_c = F(4), F_d = F(3), F_e = F(2), F_f = F(1), and F_g = F(0).


```
entity decoder_7seg_4in is
    port (A      : in Bit_vector(3 downto 0);
          F      : out bit_vector(6 downto 0));
end entity;
```

Fig. 6.4
7-Segment Display Decoder Entity

6.1.7 Design a VHDL model for a 4-input, seven-segment HEX character decoder using selected signal assignments. Use the entity definition given in Fig. 6.4 for your design. The system has a 4-bit input vector called A and a 7-bit output vector called F. The individual scalars within the output vector (i.e., F (6 downto 0)) correspond to the character display segments a, b, c, d, e, f, and g, respectively. A logic 1 on an output corresponds to the LED being ON. The display will show the HEX characters 0–9, A, b, c, d, E, and F corresponding to the 4-bit input code on A. A template for creating the truth table for this system is provided in Fig. 6.3. The signals in this table correspond to the entity in this problem as follows: A = A(3), B = A(2), C = A(1), D = A(0), F_a = F(6), F_b = F(5), F_c = F(4), F_d = F(3), F_e = F(2), F_f = F(1), and F_g = F(0).

Section 6.2—Encoders

6.2.1 Design an 8-to-3 binary encoder by hand. The block diagram and truth table for the encoder are given in Fig. 6.5. Give the logic expressions for each output and the full logic diagram for the system.



A(7)	A(6)	A(5)	A(4)	A(3)	A(2)	A(1)	A(0)	F(2)	F(1)	F(0)
0	0	0	0	0	0	0	1	0	0	0
0	0	0	0	0	0	1	0	0	0	1
0	0	0	0	0	1	0	0	0	1	0
0	0	0	0	1	0	0	0	0	1	1
0	0	0	1	0	0	0	0	1	0	0
0	0	1	0	0	0	0	0	1	0	1
0	1	0	0	0	0	0	0	1	1	0
1	0	0	0	0	0	0	0	1	1	1

Fig. 6.5
8-to-3 One-Hot Encoder Functionality

6.2.2 Design a VHDL model for an 8-to-3 binary encoder using concurrent signal assignments and logical operators. Use the entity definition given in Fig. 6.6 for your design.

```
entity encoder_8to3_binary is
  port (A : in Bit_vector (7 downto 0);
        F : out bit_vector (2 downto 0));
end entity;
```

Fig. 6.6
8-to-3 One-Hot Encoder Entity

6.2.3 Design a VHDL model for an 8-to-3 binary encoder using conditional signal assignments. Use the entity definition given in Fig. 6.6 for your design.

6.2.4 Design a VHDL model for an 8-to-3 binary encoder using selected signal assignments. Use the entity definition given in Fig. 6.6 for your design.

Section 6.3—Multiplexers

6.3.1 Design an 8-to-1 multiplexer by hand. The block diagram and truth table for the multiplexer are given in Fig. 6.7. Give the minimized logic expressions for the output and the full logic diagram for the system.

8-to-1 Multiplexer




Fig. 6.7
8-to-1 Multiplexer Functionality

6.3.2 Design a VHDL model for an 8-to-1 multiplexer using concurrent signal assignments and logical operators. Use the entity definition given in Fig. 6.8 for your design.

```
entity mux_8to1 is
  port (A : in bit_vector (7 downto 0);
        Sel : in bit_vector (2 downto 0);
        F : out bit);
end entity;
```


Fig. 6.8
8-to-1 Multiplexer Entity

6.3.3 Design a VHDL model for an 8-to-1 multiplexer using conditional signal assignments. Use the entity definition given in Fig. 6.8 for your design.

6.3.4 Design a VHDL model for an 8-to-1 multiplexer using selected signal assignments. Use the entity definition given in Fig. 6.8 for your design.

Section 6.4—Demultiplexers

6.4.1 Design a 1-to-8 demultiplexer by hand. The block diagram and truth table for the demultiplexer are given in Fig. 6.9. Give the minimized logic expressions for each output and the full logic diagram for the system.



Sel ₂	Sel ₁	Sel ₀	F ₇	F ₆	F ₅	F ₄	F ₃	F ₂	F ₁	F ₀
0	0	0	0	0	0	0	0	0	0	A
0	0	1	0	0	0	0	0	0	A	0
0	1	0	0	0	0	0	0	A	0	0
0	1	1	0	0	0	0	A	0	0	0
1	0	0	0	0	0	A	0	0	0	0
1	0	1	0	0	A	0	0	0	0	0
1	1	0	0	A	0	0	0	0	0	0
1	1	1	A	0	0	0	0	0	0	0

Fig. 6.9
1-to-8 Demultiplexer Functionality

6.4.2 Design a VHDL model for a 1-to-8 demultiplexer using concurrent signal assignments and logical operators. Use the entity definition given in Fig. 6.10 for your design.

```
entity demux_1to8 is
  port (A : in bit;
        Sel : in bit_vector (2 downto 0);
        F : out bit_vector (7 downto 0));
end entity;
```

Fig. 6.10
1-to-8 Demultiplexer Entity

6.4.3 Design a VHDL model for a 1-to-8 demultiplexer using conditional signal assignments. Use the entity definition given in Fig. 6.10 for your design.

6.4.4 Design a VHDL model for a 1-to-8 demultiplexer using selected signal assignments. Use the entity definition given in Fig. 6.10 for your design.